
Specification for the FIRRTL Language

Patrick S. Li
Adam M. Izraelevitz
Jonathan Bachrach

Electrical Engineering and Computer Sciences
University of California at Berkeley

Technical Report No. UCB/EECS-2016-9

http://www.eecs.berkeley.edu/Pubs/TechRpts/2016/EECS-2016-9.html

February 24, 2016



Copyright © 2016, by the author(s).
All rights reserved.

 
Permission to make digital or hard copies of all or part of this work for
personal or classroom use is granted without fee provided that copies are
not made or distributed for profit or commercial advantage and that copies
bear this notice and the full citation on the first page. To copy otherwise, to
republish, to post on servers or to redistribute to lists, requires prior specific
permission.



Specification for the FIRRTL Language

Patrick S. Li
psli@eecs.berkeley.edu

Adam M. Izraelevitz
adamiz@eecs.berkeley.edu

Jonathan Bachrach
jrb@eecs.berkeley.edu

February 23, 2016

Contents

1 Introduction 5
1.1 Background . . . . . . . . . . . . . . . . . . . . . . . . . . . . 5
1.2 Design Philosophy . . . . . . . . . . . . . . . . . . . . . . . . 6

2 Acknowledgements 6

3 Circuits and Modules 8
3.1 Circuits . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 8
3.2 Modules . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 8
3.3 Externally Defined Modules . . . . . . . . . . . . . . . . . . . 9

4 Types 9
4.1 Ground Types . . . . . . . . . . . . . . . . . . . . . . . . . . . 9

4.1.1 Integer Types . . . . . . . . . . . . . . . . . . . . . . . 9
4.1.2 Clock Type . . . . . . . . . . . . . . . . . . . . . . . . 10

4.2 Vector Types . . . . . . . . . . . . . . . . . . . . . . . . . . . 10
4.3 Bundle Types . . . . . . . . . . . . . . . . . . . . . . . . . . . 10
4.4 Passive Types . . . . . . . . . . . . . . . . . . . . . . . . . . . 11
4.5 Type Equivalence . . . . . . . . . . . . . . . . . . . . . . . . . 12
4.6 Weak Type Equivalence . . . . . . . . . . . . . . . . . . . . . 12

4.6.1 Oriented Types . . . . . . . . . . . . . . . . . . . . . . 12

1

mailto:psli@eecs.berkeley.edu
mailto:adamiz@eecs.berkeley.edu
mailto:jrb@eecs.berkeley.edu


Specification for the FIRRTL Language Version 0.2.0

4.6.2 Conversion to Oriented Types . . . . . . . . . . . . . . 13
4.6.3 Oriented Type Equivalence . . . . . . . . . . . . . . . . 13

5 Statements 13
5.1 Connects . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 13

5.1.1 The Connection Algorithm . . . . . . . . . . . . . . . . 14
5.2 Partial Connects . . . . . . . . . . . . . . . . . . . . . . . . . 15

5.2.1 The Partial Connection Algorithm . . . . . . . . . . . 16
5.3 Statement Groups . . . . . . . . . . . . . . . . . . . . . . . . . 16

5.3.1 Last Connect Semantics . . . . . . . . . . . . . . . . . 17
5.4 Empty . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 18
5.5 Wires . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 19
5.6 Registers . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 19
5.7 Invalidates . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 19

5.7.1 The Invalidate Algorithm . . . . . . . . . . . . . . . . 20
5.8 Nodes . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 21
5.9 Conditionals . . . . . . . . . . . . . . . . . . . . . . . . . . . . 21

5.9.1 Syntactic Shorthands . . . . . . . . . . . . . . . . . . . 21
5.9.2 Nested Declarations . . . . . . . . . . . . . . . . . . . . 23
5.9.3 Initialization Coverage . . . . . . . . . . . . . . . . . . 24
5.9.4 Scoping . . . . . . . . . . . . . . . . . . . . . . . . . . 24
5.9.5 Conditional Last Connect Semantics . . . . . . . . . . 25

5.10 Memories . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 27
5.10.1 Read Ports . . . . . . . . . . . . . . . . . . . . . . . . 28
5.10.2 Write Ports . . . . . . . . . . . . . . . . . . . . . . . . 29
5.10.3 Readwrite Ports . . . . . . . . . . . . . . . . . . . . . . 29
5.10.4 Read Under Write Behaviour . . . . . . . . . . . . . . 30

5.11 Instances . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 30
5.12 Stops . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 31
5.13 Formatted Prints . . . . . . . . . . . . . . . . . . . . . . . . . 31

5.13.1 Format Strings . . . . . . . . . . . . . . . . . . . . . . 32

6 Expressions 32
6.1 Unsigned Integers . . . . . . . . . . . . . . . . . . . . . . . . . 32
6.2 Unsigned Integers from Literal Bits . . . . . . . . . . . . . . . 33
6.3 Signed Integers . . . . . . . . . . . . . . . . . . . . . . . . . . 34
6.4 Signed Integers from Literal Bits . . . . . . . . . . . . . . . . 34
6.5 References . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 35

2



Specification for the FIRRTL Language Version 0.2.0

6.6 Subfields . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 35
6.7 Subindices . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 35
6.8 Subaccesses . . . . . . . . . . . . . . . . . . . . . . . . . . . . 36
6.9 Multiplexors . . . . . . . . . . . . . . . . . . . . . . . . . . . . 38
6.10 Conditionally Valids . . . . . . . . . . . . . . . . . . . . . . . 39
6.11 Primitive Operations . . . . . . . . . . . . . . . . . . . . . . . 39

7 Primitive Operations 40
7.1 Add Operation . . . . . . . . . . . . . . . . . . . . . . . . . . 40
7.2 Subtract Operation . . . . . . . . . . . . . . . . . . . . . . . . 40
7.3 Multiply Operation . . . . . . . . . . . . . . . . . . . . . . . . 41
7.4 Divide Operation . . . . . . . . . . . . . . . . . . . . . . . . . 41
7.5 Modulus Operation . . . . . . . . . . . . . . . . . . . . . . . . 41
7.6 Comparison Operations . . . . . . . . . . . . . . . . . . . . . . 42
7.7 Padding Operations . . . . . . . . . . . . . . . . . . . . . . . . 42
7.8 Interpret As UInt . . . . . . . . . . . . . . . . . . . . . . . . . 42
7.9 Interpret As SInt . . . . . . . . . . . . . . . . . . . . . . . . . 43
7.10 Interpret as Clock . . . . . . . . . . . . . . . . . . . . . . . . . 43
7.11 Shift Left Operation . . . . . . . . . . . . . . . . . . . . . . . 43
7.12 Shift Right Operation . . . . . . . . . . . . . . . . . . . . . . . 43
7.13 Dynamic Shift Left Operation . . . . . . . . . . . . . . . . . . 44
7.14 Dynamic Shift Right Operation . . . . . . . . . . . . . . . . . 44
7.15 Arithmetic Convert to Signed Operation . . . . . . . . . . . . 44
7.16 Negate Operation . . . . . . . . . . . . . . . . . . . . . . . . . 44
7.17 Bitwise Complement Operation . . . . . . . . . . . . . . . . . 45
7.18 Binary Bitwise Operations . . . . . . . . . . . . . . . . . . . . 45
7.19 Bitwise Reduction Operations . . . . . . . . . . . . . . . . . . 45
7.20 Concatenate Operation . . . . . . . . . . . . . . . . . . . . . . 45
7.21 Bit Extraction Operation . . . . . . . . . . . . . . . . . . . . . 46
7.22 Head . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 46
7.23 Tail . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 46

8 Genders 46

9 Width Inference 47

3



Specification for the FIRRTL Language Version 0.2.0

10 Namespaces 48
10.1 Name Expansion Algorithm . . . . . . . . . . . . . . . . . . . 48
10.2 Prefix Uniqueness . . . . . . . . . . . . . . . . . . . . . . . . . 48

11 The Lowered FIRRTL Form 49

12 Details about Syntax 50

13 FIRRTL Language Definition 52
13.1 Notation . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 52
13.2 Concrete Syntax Tree . . . . . . . . . . . . . . . . . . . . . . . 53

4



Specification for the FIRRTL Language Version 0.2.0

1 Introduction

1.1 Background

The ideas for FIRRTL (Flexible Intermediate Representation for RTL) orig-
inated from work on Chisel, a hardware description language (HDL) embed-
ded in Scala used for writing highly-parameterized circuit design generators.
Chisel designers manipulate circuit components using Scala functions, encode
their interfaces in Scala types, and use Scala’s object-orientation features to
write their own circuit libraries. This form of meta-programming enables
expressive, reliable and type-safe generators that improve RTL design pro-
ductivity and robustness.

The computer architecture research group at U.C. Berkeley relies criti-
cally on Chisel to allow small teams of graduate students to design sophis-
ticated RTL circuits. Over a three year period with under twelve graduate
students, the architecture group has taped-out over ten different designs.

Internally, the investment in developing and learning Chisel was rewarded
with huge gains in productivity. However, Chisel’s external rate of adoption
was slow for the following reasons.

1. Writing custom circuit transformers requires intimate knowledge about
the internals of the Chisel compiler.

2. Chisel semantics are underspecified and thus impossible to target from
other languages.

3. Error checking is unprincipled due to underspecified semantics resulting
in incomprehensible error messages.

4. Learning a functional programming language (Scala) is difficult for RTL
designers with limited programming language experience.

5. Confounding the previous point, conceptually separating the embedded
Chisel HDL from the host language is difficult for new users.

6. The output of Chisel (Verilog) is unreadable and slow to simulate.

As a consequence, Chisel needed to be redesigned from the ground up
to standardize its semantics, modularize its compilation process, and cleanly
separate its front-end, intermediate representation, and backends. A well
defined intermediate representation (IR) allows the system to be targeted
by other HDLs embedded in other host programming languages, making it
possible for RTL designers to work within a language they are already com-
fortable with. A clearly defined IR with a concrete syntax also allows for

inspection of the output of circuit generators and transformers thus making
clear the distinction between the host language and the constructed circuit.
Clearly defined semantics allows users without knowledge of the compiler
implementation to write circuit transformers; examples include optimization
of circuits for simulation speed, and automatic insertion of signal activity
counters. An additional benefit of a well defined IR is the structural invari-
ants that can be enforced before and after each compilation stage, resulting
in a more robust compiler and structured mechanism for error checking.

1.2 Design Philosophy

FIRRTL represents the standardized elaborated circuit that the Chisel HDL
produces. FIRRTL represents the circuit immediately after Chisel’s elabora-
tion but before any circuit simplification. It is designed to resemble the Chisel
HDL after all meta-programming has executed. Thus, a user program that
makes little use of meta-programming facilities should look almost identical
to the generated FIRRTL.

For this reason, FIRRTL has first-class support for high-level constructs
such as vector types, bundle types, conditional statements, partial connects,
and modules. These high-level constructs are then gradually removed by a
sequence of lowering transformations. During each lowering transformation,
the circuit is rewritten into an equivalent circuit using simpler, lower-level
constructs. Eventually the circuit is simplified to its most restricted form,
resembling a structured netlist, which allows for easy translation to an out-
put language (e.g. Verilog). This form is given the name lowered FIRRTL
(LoFIRRTL) and is a strict subset of the full FIRRTL language.

Because the host language is now used solely for its meta-programming
facilities, the frontend can be very light-weight, and additional HDLs written
in other languages can target FIRRTL and reuse the majority of the compiler
toolchain.

2 Acknowledgements

The FIRRTL language could not have been developed without the help of
many of the faculty and students in the ASPIRE lab, and the University of
California, Berkeley.

This project originated from discussions with the authors’ advisor, Jonathan

Bachrach, who indicated the need for a structural redesign of the Chisel sys-
tem around a well-defined intermediate representation. Patrick Li designed
and implemented the first prototype of the FIRRTL language, wrote the ini-
tial specification for the language, and presented it to the Chisel group con-
sisting of Adam Izraelevitz, Scott Beamer, David Biancolin, Christopher Ce-
lio, Henry Cook, Palmer Dabbelt, Donggyu Kim, Jack Koenig, Martin Maas,
Albert Magyar, Colin Schmidt, Andrew Waterman, Yunsup Lee, Richard
Lin, Eric Love, Albert Ou, Stephen Twigg, John Bachan, David Donofrio,
Farzad Fatollahi-Fard, Jim Lawson, Brian Richards, Krste Asanović, and
John Wawrzynek.

Adam Izraelevitz then reworked the design and reimplemented FIRRTL,
and after many discussions with Patrick Li and the Chisel group, refined the
design to its present version.

The authors would like to thank the following individuals in particular
for their contributions to the FIRRTL project:

• Andrew Waterman: for his many contributions to the design of FIR-
RTL’s constructs, for his work on Chisel 3.0, and for porting architec-
ture research infrastructure

• Richard Lin: for improving the Chisel 3.0 code base for release quality

• Jack Koenig: for implementing the FIRRTL parser in Scala

• Henry Cook: for porting and cleaning up many aspects of Chisel 3.0,
including the testing infrastructure and the parameterization library

• Chick Markley: for creating the new testing harness and porting the
Chisel tutorial

• Stephen Twigg: for his expertise in hardware intermediate representa-
tions and for providing many corner cases to consider

• Palmer Dabbelt, Eric Love, Martin Maas, Christopher Celio, and Scott
Beamer: for their feedback on previous drafts of the FIRRTL specifi-
cation

And finally this project would not have been possible without the contin-
uous feedback and encouragement of Jonathan Bachrach, and his leadership
on and implementation of Chisel.

This research was partially funded by DARPA Award Number HR0011-
12-2-0016, the Center for Future Architecture Research, a member of STAR-
net, a Semiconductor Research Corporation program sponsored by MARCO
and DARPA, and ASPIRE Lab industrial sponsors and affiliates Intel, Google,
Huawei, Nokia, NVIDIA, Oracle, and Samsung. Any opinions, findings, con-
clusions, or recommendations in this paper are solely those of the authors
and does not necessarily reflect the position or the policy of the sponsors.

3 Circuits and Modules

3.1 Circuits

All FIRRTL circuits consist of a list of modules, each representing a hardware
block that can be instantiated. The circuit must specify the name of the top-
level module.

circuit MyTop :

module MyTop :

...

module MyModule :

...

3.2 Modules

Each module has a given name, a list of ports, and a statement representing
the circuit connections within the module. A module port is specified by its
direction, which may be input or output, a name, and the data type of the
port.

The following example declares a module with one input port, one output
port, and one statement connecting the input port to the output port. See
section 5.1 for details on the connect statement.

module MyModule :

input foo: UInt

output bar: UInt

bar <= foo

Note that a module definition does not indicate that the module will be
physically present in the final circuit. Refer to the description of the instance
statement for details on how to instantiate a module (section 5.11).

3.3 Externally Defined Modules

Externally defined modules consist of a given name, and a list of ports, whose
types and names must match its external definition.

module MyExternalModule :

input foo: UInt

output bar: UInt

output baz: SInt

4 Types

Types are used to specify the structure of the data held by each circuit
component. All types in FIRRTL are either one of the fundamental ground
types or are built up from aggregating other types.

4.1 Ground Types

There are three ground types in FIRRTL: an unsigned integer type, a signed
integer type, and a clock type.

4.1.1 Integer Types

Both unsigned and signed integer types may optionally be given a known
positive integer bit width.

UInt<10>

SInt<32>

Alternatively, if the bit width is omitted, it will be automatically inferred
by FIRRTL’s width inferencer, as detailed in section 9.

UInt

SInt

4.1.2 Clock Type

The clock type is used to describe wires and ports meant for carrying clock
signals. The usage of components with clock types are restricted. Clock
signals cannot be used in most primitive operations, and clock signals can
only be connected to components that have been declared with the clock
type.

The clock type is specified as follows:

Clock

4.2 Vector Types

A vector type is used to express an ordered sequence of elements of a given
type. The length of the sequence must be non-negative and known.

The following example specifies a ten element vector of 16-bit unsigned
integers.

UInt<16>[10]

The next example specifies a ten element vector of unsigned integers of
omitted but identical bit widths.

UInt[10]

Note that any type, including other aggregate types, may be used as the
element type of the vector. The following example specifies a twenty element
vector, each of which is a ten element vector of 16-bit unsigned integers.

UInt<16>[10][20]

4.3 Bundle Types

A bundle type is used to express a collection of nested and named types. All
fields in a bundle type must have a given name, and type.

The following is an example of a possible type for representing a complex
number. It has two fields, real, and imag, both 10-bit signed integers.

{real:SInt<10>, imag:SInt<10>}

Additionally, a field may optionally be declared with a flipped orientation.

{word:UInt<32>, valid:UInt<1>, flip ready:UInt<1>}

In a connection between circuit components with bundle types, the data
carried by the flipped fields flow in the opposite direction as the data carried
by the non-flipped fields.

As an example, consider a module output port declared with the following
type:

output a: {word:UInt<32>, valid:UInt<1>, flip ready:UInt<1>}

In a connection to the a port, the data carried by the word and valid

subfields will flow out of the module, while data carried by the ready subfield
will flow into the module. More details about how the bundle field orientation
affects connections are explained in section 5.1.

As in the case of vector types, a bundle field may be declared with any
type, including other aggregate types.

{real: {word:UInt<32>, valid:UInt<1>, flip ready:UInt<1>}

imag: {word:UInt<32>, valid:UInt<1>, flip ready:UInt<1>}}

When calculating the final direction of data flow, the orientation of a field
is applied recursively to all nested types in the field. As an example, consider
the following module port declared with a bundle type containing a nested
bundle type.

output myport: {a: UInt, flip b: {c: UInt, flip d:UInt}}

In a connection to myport, the a subfield flows out of the module. The c

subfield contained in the b subfield flows into the module, and the d subfield
contained in the b subfield flows out of the module.

4.4 Passive Types

It is inappropriate for some circuit components to be declared with a type
that allows for data to flow in both directions. For example, all subelements
in a memory should flow in the same direction. These components are re-
stricted to only have a passive type.

Intuitively, a passive type is a type where all data flows in the same
direction, and is defined to be a type that recursively contains no fields with
flipped orientations. Thus all ground types are passive types. Vector types
are passive if their element type is passive. And bundle types are passive if
no fields are flipped and if all field types are passive.

4.5 Type Equivalence

The type equivalence relation is used to determine whether a connection
between two components is legal. See section 5.1 for further details about
connect statements.

An unsigned integer type is always equivalent to another unsigned integer
type regardless of bit width, and is not equivalent to any other type. Sim-
ilarly, a signed integer type is always equivalent to another signed integer
type regardless of bit width, and is not equivalent to any other type.

Clock types are equivalent to clock types, and are not equivalent to any
other type.

Two vector types are equivalent if they have the same length, and if their
element types are equivalent.

Two bundle types are equivalent if they have the same number of fields,
and both the bundles’ i’th fields have matching names and orientations, as
well as equivalent types. Consequently, {a:UInt, b:UInt} is not equiva-
lent to {b:UInt, a:UInt}, and {a: {flip b:UInt}} is not equivalent to
{flip a: {b: UInt}}.

4.6 Weak Type Equivalence

The weak type equivalence relation is used to determine whether a partial
connection between two components is legal. See section 5.2 for further
details about partial connect statements.

Two types are weakly equivalent if their corresponding oriented types are
equivalent.

4.6.1 Oriented Types

The weak type equivalence relation requires first a definition of oriented types.
Intuitively, an oriented type is a type where all orientation information is
collated and coupled with the leaf ground types instead of in bundle fields.

An oriented ground type is an orientation coupled with a ground type.
An oriented vector type is an ordered sequence of positive length of elements
of a given oriented type. An oriented bundle type is a collection of oriented
fields, each containing a name and an oriented type, but no orientation.

Applying a flip orientation to an oriented type recursively reverses the
orientation of every oriented ground type contained within. Applying a non-

12



Specification for the FIRRTL Language Version 0.2.0

flip orientation to an oriented type does nothing.

4.6.2 Conversion to Oriented Types

To convert a ground type to an oriented ground type, attach a non-flip ori-
entation to the ground type.

To convert a vector type to an oriented vector type, convert its element
type to an oriented type, and retain its length.

To convert a bundle field to an oriented field, convert its type to an
oriented type, apply the field orientation, and combine this with the original
field’s name to create the oriented field. To convert a bundle type to an
oriented bundle type, convert each field to an oriented field.

4.6.3 Oriented Type Equivalence

Two oriented ground types are equivalent if their orientations match and
their types are equivalent.

Two oriented vector types are equivalent if their element types are equiv-
alent.

Two oriented bundle types are not equivalent if there exists two fields,
one from each oriented bundle type, that have identical names but whose
oriented types are not equivalent. Otherwise, the oriented bundle types are
equivalent.

As stated earlier, two types are weakly equivalent if their corresponding
oriented types are equivalent.

5 Statements

Statements are used to describe the components within a module and how
they interact.

5.1 Connects

The connect statement is used to specify a physically wired connection be-
tween two circuit components.

The following example demonstrates connecting a module’s input port to
its output port, where port myinput is connected to port myoutput.

13

module MyModule :

input myinput: UInt

output myoutput: UInt

myoutput <= myinput

In order for a connection to be legal the following conditions must hold:

1. The types of the left-hand and right-hand side expressions must be
equivalent (see section 4.5 for details).

2. The bit widths of the two expressions must allow for data to always
flow from a smaller bit width to an equal size or larger bit width.

3. The gender of the left-hand side expression must be female or bi-gender
(see section 8 for an explanation of gender).

4. Either the gender of the right-hand side expression is male or bi-gender,
or the right-hand side expression has a passive type.

Connect statements from a narrower ground type component to a wider
ground type component will have its value automatically sign-extended or
zero-extended to the larger bit width. The behaviour of connect statements
between two circuit components with aggregate types is defined by the con-
nection algorithm in section 5.1.1.

5.1.1 The Connection Algorithm

Connect statements between ground types cannot be expanded further.
Connect statements between two vector typed components recursively

connects each subelement in the right-hand side expression to the corre-
sponding subelement in the left-hand side expression.

Connect statements between two bundle typed components connects the
i’th field of the right-hand side expression and the i’th field of the left-hand
side expression. If the i’th field is not flipped, then the right-hand side field
is connected to the left-hand side field. Conversely, if the i’th field is flipped,
then the left-hand side field is connected to the right-hand side field.

14

5.2 Partial Connects

Like the connect statement, the partial connect statement is also used to
specify a physically wired connection between two circuit components. How-
ever, it enforces fewer restrictions on the types and widths of the circuit
components it connects.

In order for a partial connect to be legal the following conditions must
hold:

1. The types of the left-hand and right-hand side expressions must be
weakly equivalent (see section 4.6 for details).

2. The gender of the left-hand side expression must be female or bi-gender
(see section 8 for an explanation of gender).

3. Either the gender of the right-hand side expression is male or bi-gender,
or the right-hand side expression has a passive type.

Partial connect statements from a narrower ground type component to a
wider ground type component will have its value automatically sign-extended
to the larger bit width. Partial connect statements from a wider ground
type component to a narrower ground type component will have its value
automatically truncated to fit the smaller bit width.

Intuitively, bundle fields with matching names will be connected appropri-
ately, while bundle fields not present in both types will be ignored. Similarly,
vectors with mismatched lengths will be connected up to the shorter length,
and the remaining subelements are ignored. The full algorithm is detailed in
section 5.2.1.

The following example demonstrates partially connecting a module’s in-
put port to its output port, where port myinput is connected to port myoutput.

module MyModule :

input myinput: {flip a:UInt, b:UInt[2]}

output myoutput: {flip a:UInt, b:UInt[3], c:UInt}

myoutput <- myinput

The above example is equivalent to the following:

module MyModule :

input myinput: {flip a:UInt, b:UInt[2]}

output myoutput: {flip a:UInt, b:UInt[3], c:UInt}

myinput.a <- myoutput.a

myoutput.b[0] <- myinput.b[0]

myoutput.b[1] <- myinput.b[1]

For details on the syntax and semantics of the subfield expression, subindex
expression, and statement groups, see sections 6.6, 6.7, and 5.3.

5.2.1 The Partial Connection Algorithm

A partial connect statement between two ground type components connects
the right-hand side expression to the left-hand side expression. Conversely,
a reverse partial connect statement between two ground type components
connects the left-hand side expression to the right-hand side expression.

A partial (or reverse partial) connect statement between two vector typed
components applies a partial (or reverse partial) connect from the first n
subelements in the right-hand side expression to the first n corresponding
subelements in the left-hand side expression, where n is the length of the
shorter vector.

A partial (or reverse partial) connect statement between two bundle typed
components considers any pair of fields, one from the first bundle type and
one from the second, with matching names. If the first field in the pair is not
flipped, then we apply a partial (or reverse partial) connect from the right-
hand side field to the left-hand side field. However, if the first field is flipped,
then we apply a reverse partial (or partial) connect from the right-hand side
field to the left-hand side field.

5.3 Statement Groups

An ordered sequence of one or more statements can be grouped into a single
statement, called a statement group. The following example demonstrates a
statement group composed of three connect statements.

module MyModule :

input a: UInt

input b: UInt

output myport1: UInt

output myport2: UInt

myport1 <= a

myport1 <= b

myport2 <= a

5.3.1 Last Connect Semantics

Ordering of statements is significant in a statement group. Intuitively, during
elaboration, statements execute in order, and the effects of later statements
take precedence over earlier ones. In the previous example, in the resultant
circuit, port b will be connected to myport1, and port a will be connected to
myport2.

Note that connect and partial connect statements have equal priority,
and later connect or partial connect statements always take priority over
earlier connect or partial connect statements. Conditional statements are
also affected by last connect semantics, and for details see section 5.9.5.

In the case where a connection to a circuit component with an aggregate
type is followed by a connection to a subelement of that component, only
the connection to the subelement is overwritten. Connections to the other
subelements remain unaffected. In the following example, in the resultant
circuit, the c subelement of port portx will be connected to the c subelement
of myport, and port porty will be connected to the b subelement of myport.

module MyModule :

input portx: {b:UInt, c:UInt}

input porty: UInt

output myport: {b:UInt, c:UInt}

myport <= portx

myport.b <= porty

The above circuit can be rewritten equivalently as follows.

module MyModule :

input portx: {b:UInt, c:UInt}

input porty: UInt

output myport: {b:UInt, c:UInt}

myport.b <= porty

myport.c <= portx.c

In the case where a connection to a subelement of an aggregate circuit
component is followed by a connection to the entire circuit component, the
later connection overwrites the earlier connections completely.

module MyModule :

input portx: {b:UInt, c:UInt}

input porty: UInt

output myport: {b:UInt, c:UInt}

myport.b <= porty

myport <= portx

The above circuit can be rewritten equivalently as follows.

module MyModule :

input portx: {b:UInt, c:UInt}

input porty: UInt

output myport: {b:UInt, c:UInt}

myport <= portx

See section 6.6 for more details about subfield expressions.

5.4 Empty

The empty statement does nothing and is used simply as a placeholder where
a statement is expected. It is specified using the skip keyword.

The following example:

a <= b

skip

c <= d

can be equivalently expressed as:

a <= b

c <= d

The empty statement is most often used as the else branch in a con-
ditional statement, or as a convenient placeholder for removed components
during transformational passes. See section 5.9 for details on the conditional
statement.

5.5 Wires

A wire is a named combinational circuit component that can be connected
to and from using connect and partial connect statements.

The following example demonstrates instantiating a wire with the given
name mywire and type UInt.

wire mywire : UInt

5.6 Registers

A register is a named stateful circuit component.
The following example demonstrates instantiating a register with the

given name myreg, type SInt, and is driven by the clock signal myclock.

wire myclock: Clock

reg myreg: SInt, myclock

...

Optionally, for the purposes of circuit initialization, a register can be
declared with a reset signal and value. In the following example, myreg is
assigned the value myinit when the signal myreset is high.

wire myclock: Clock

wire myreset: UInt<1>

wire myinit: SInt

reg myreg: SInt, myclock, myreset, myinit

...

Note that the clock signal for a register must be of type clock, the reset
signal must be a single bit UInt, and the type of initialization value must
match the declared type of the register.

5.7 Invalidates

An invalidate statement is used to indicate that a circuit component contains
indeterminate values. It is specified as follows:

wire w:UInt

w is invalid

Invalidate statements can be applied to any circuit component of any
type. However, if the circuit component cannot be connected to, then the
statement has no effect on the component. This allows the invalidate state-
ment to be applied to any component, to explicitly ignore initialization cov-
erage errors.

The following example demonstrates the effect of invalidating a variety
of circuit components with aggregate types. See section 5.7.1 for details on
the algorithm for determining what is invalidated.

module MyModule :

input in: {flip a:UInt, b:UInt}

output out: {flip a:UInt, b:UInt}

wire w: {flip a:UInt, b:UInt}

in is invalid

out is invalid

w is invalid

is equivalent to the following:

module MyModule :

input in: {flip a:UInt, b:UInt}

output out: {flip a:UInt, b:UInt}

wire w: {flip a:UInt, b:UInt}

in.a is invalid

out.b is invalid

w.a is invalid

w.b is invalid

For the purposes of simulation, invalidated components are initialized
to random values, and operations involving indeterminate values produce
undefined behaviour. This is useful for early detection of errors in simulation.

5.7.1 The Invalidate Algorithm

Invalidating a component with a ground type indicates that the component’s
value is indetermined if the component is female or bi-gender (see section 8).
Otherwise, the component is unaffected.

Invalidating a component with a vector type recursively invalidates each
subelement in the vector.

Invalidating a component with a bundle type recursively invalidates each
subelement in the bundle.

5.8 Nodes

A node is simply a named intermediate value in a circuit. The node must
be initialized to a value with a passive type and cannot be connected to.
Nodes are often used to split a complicated compound expression into named
subexpressions.

The following example demonstrates instantiating a node with the given
name mynode initialized with the output of a multiplexor (see section 6.9).

wire pred: UInt<1>

wire a: SInt

wire b: SInt

node mynode = mux(pred, a, b)

...

5.9 Conditionals

Connections within a conditional statement that connect to previously de-
clared components hold only when the given condition is high. The condition
must have a 1-bit unsigned integer type.

In the following example, the wire x is connected to the input a only when
the en signal is high. Otherwise, the wire x is connected to the input b.

module MyModule :

input a: UInt

input b: UInt

input en: UInt<1>

wire x: UInt

when en :

x <= a

else :

x <= b

5.9.1 Syntactic Shorthands

The else branch of a conditional statement may be omitted, in which case
a default else branch is supplied consisting of the empty statement.

Thus the following example:

module MyModule :

input a: UInt

input b: UInt

input en: UInt<1>

wire x: UInt

when en :

x <= a

can be equivalently expressed as:

module MyModule :

input a: UInt

input b: UInt

input en: UInt<1>

wire x: UInt

when en :

x <= a

else :

skip

To aid readability of long chains of conditional statements, the colon
following the else keyword may be omitted if the else branch consists of a
single conditional statement.

Thus the following example:

module MyModule :

input a: UInt

input b: UInt

input c: UInt

input d: UInt

input c1: UInt<1>

input c2: UInt<1>

input c3: UInt<1>

wire x: UInt

when c1 :

x <= a

else :

when c2 :

x <= b

else :

when c3 :

x <= c

else :

x <= d

can be equivalently written as:

module MyModule :

input a: UInt

input b: UInt

input c: UInt

input d: UInt

input c1: UInt<1>

input c2: UInt<1>

input c3: UInt<1>

wire x: UInt

when c1 :

x <= a

else when c2 :

x <= b

else when c3 :

x <= c

else :

x <= d

5.9.2 Nested Declarations

If a component is declared within a conditional statement, connections to the
component are unaffected by the condition. In the following example, register
myreg1 is always connected to a, and register myreg2 is always connected to
b.

module MyModule :

input a: UInt

input b: UInt

input en: UInt<1>

input clk : Clock

when en :

reg myreg1 : UInt, clk

myreg1 <= a

else :

reg myreg2 : UInt, clk

myreg2 <= b

Intuitively, a line can be drawn between a connection (or partial con-
nection) to a component and that component’s declaration. All conditional
statements that are crossed by the line apply to that connection (or partial
connection).

5.9.3 Initialization Coverage

Because of the conditional statement, it is possible to syntactically express
circuits containing wires that have not been connected to under all conditions.

In the following example, the wire a is connected to the wire w when en

is high, but it is not specified what is connected to w when en is low.

module MyModule :

input en: UInt<1>

input a: UInt

wire w: UInt

when en :

w <= a

This is an illegal FIRRTL circuit and an error will be thrown during
compilation. All wires, memory ports, instance ports, and module ports that
can be connected to must be connected to under all conditions. Registers do
not need to be connected to under all conditions, as it will keep its previous
value if unconnected.

5.9.4 Scoping

The conditional statement creates a new scope within each of its when and
else branches. It is an error to refer to any component declared within a
branch after the branch has ended.

5.9.5 Conditional Last Connect Semantics

In the case where a connection to a circuit component is followed by a condi-
tional statement containing a connection to the same component, the connec-
tion is overwritten only when the condition holds. Intuitively, a multiplexor
is generated such that when the condition is low, the multiplexor returns
the old value, and otherwise returns the new value. For details about the
multiplexor, see section 6.9.

The following example:

wire a: UInt

wire b: UInt

wire c: UInt<1>

wire w: UInt

w <= a

when c :

w <= b

...

can be rewritten equivalently using a multiplexor as follows:

wire a: UInt

wire b: UInt

wire c: UInt<1>

wire w: UInt

w <= mux(c, b, a)

...

In the case where an invalid statement is followed by a conditional state-
ment containing a connect to the invalidated component, the resulting con-
nection to the component can be expressed using a conditionally valid ex-
pression. See section 6.10 for more details about the conditionally valid
expression.

wire a: UInt

wire c: UInt<1>

wire w: UInt

w is invalid

when c :

w <= a

...

can be rewritten equivalently as follows:

wire a: UInt

wire c: UInt<1>

wire w: UInt

w <= validif(c, a)

...

The behaviour of conditional connections to circuit components with ag-
gregate types can be modeled by first expanding each connect into individual
connect statements on its ground elements (see section 5.1.1 and 5.2.1 for the
connection and partial connection algorithms) and then applying the condi-
tional last connect semantics.

For example, the following snippet:

wire x: {a:UInt, b:UInt}

wire y: {a:UInt, b:UInt}

wire c: UInt<1>

wire w: {a:UInt, b:UInt}

w <= x

when c :

w <= y

...

can be rewritten equivalently as follows:

wire x: {a:UInt, b:UInt}

wire y: {a:UInt, b:UInt}

wire c: UInt<1>

wire w: {a:UInt, b:UInt}

w.a <= mux(c, y.a, x.a)

w.b <= mux(c, y.b, x.b)

...

Similar to the behavior of aggregate types under last connect semantics
(see section 5.3.1), the conditional connects to a subelement of an aggregate
component only generates a multiplexor for the subelement that is overwrit-
ten.

For example, the following snippet:

wire x: {a:UInt, b:UInt}

wire y: UInt

wire c: UInt<1>

wire w: {a:UInt, b:UInt}

w <= x

when c :

w.a <= y

...

can be rewritten equivalently as follows:

wire x: {a:UInt, b:UInt}

wire y: UInt

wire c: UInt<1>

wire w: {a:UInt, b:UInt}

w.a <= mux(c, y, x.a)

w.b <= x.b

...

5.10 Memories

A memory is an abstract representation of a hardware memory. It is charac-
terized by the following parameters.

1. A passive type representing the type of each element in the memory.

2. A positive integer representing the number of elements in the memory.

3. A variable number of named ports, each being a read port, a write
port, or readwrite port.

4. A non-negative integer indicating the read latency, which is the number
of cycles after setting the port’s read address before the corresponding
element’s value can be read from the port’s data field.

5. A non-negative integer indicating the write latency, which is the num-
ber of cycles after setting the port’s write address and data before the
corresponding element within the memory holds the new value.

6. A read-under-write flag indicating the behaviour when a memory loca-
tion is written to while a read to that location is in progress.

The following example demonstrates instantiating a memory containing
256 complex numbers, each with 16-bit signed integer fields for its real and
imaginary components. It has two read ports, r1 and r2, and one write port,
w. It is combinationally read (read latency is zero cycles) and has a write
latency of one cycle. Finally, its read-under-write behavior is undefined.

mem mymem :

data-type => {real:SInt<16>, imag:SInt<16>}

depth => 256

reader => r1

reader => r2

writer => w

read-latency => 0

write-latency => 1

read-under-write => undefined

In the example above, the type of mymem is:

{flip r1: {flip data: {real:SInt<16>, imag:SInt<16>},

addr: UInt<8>,

en: UInt<1>,

clk: Clock}

flip r2: {flip data: {real:SInt<16>, imag:SInt<16>},

addr: UInt<8>,

en: UInt<1>,

clk: Clock}

flip w: {data: {real:SInt<16>, imag:SInt<16>},

mask: {real:UInt<1>, imag:UInt<1>},

addr: UInt<8>,

en: UInt<1>,

clk: Clock}}

The following sections describe how a memory’s field types are calculated
and the behavior of each type of memory port.

5.10.1 Read Ports

If a memory is declared with element type T, has a size less than or equal to
2N , then its read ports have type:

{flip data:T, addr:UInt<N>, en:UInt<1>, clk:Clock}

If the en field is high, then the element value associated with the address
in the addr field can be retrieved by reading from the data field after the
appropriate read latency. If the en field is low, then the value in the data

field, after the appropriate read latency, is undefined. The port is driven by
the clock signal in the clk field.

5.10.2 Write Ports

If a memory is declared with element type T, has a size less than or equal to
2N , then its write ports have type:

{data:T, mask:M, addr:UInt<N>, en:UInt<1>, clk:Clock}

where M is the mask type calculated from the element type T. Intuitively,
the mask type mirrors the aggregate structure of the element type except
with all ground types replaced with a single bit unsigned integer type. The
non-masked portion of the data value is defined as the set of data value leaf
subelements where the corresponding mask leaf subelement is high.

If the en field is high, then the non-masked portion of the data field value
is written, after the appropriate write latency, to the location indicated by
the addr field. If the en field is low, then no value is written after the
appropriate write latency. The port is driven by the clock signal in the clk

field.

5.10.3 Readwrite Ports

Finally, the readwrite ports have type:

{wmode:UInt<1>, flip rdata:T, data:T, mask:M,

addr:UInt<N>, en:UInt<1>, clk:Clock}

A readwrite port is a single port that, on a given cycle, can be used either
as a read or a write port. If the readwrite port is not in write mode (the
wmode field is low), then the rdata, addr, en, and clk fields constitute its
read port fields, and should be used accordingly. If the readwrite port is in
write mode (the wmode field is high), then the data, mask, addr, en, and clk

fields constitute its write port fields, and should be used accordingly.

5.10.4 Read Under Write Behaviour

The read-under-write flag indicates the value held on a read port’s data field
if its memory location is written to while it is reading. The flag may take on
three settings: old, new, and undefined.

If the read-under-write flag is set to old, then a read port always re-
turns the value existing in the memory on the same cycle that the read was
requested.

Assuming that a combinational read always returns the value stored in
the memory (no write forwarding), then intuitively, this is modeled as a
combinational read from the memory that is then delayed by the appropriate
read latency.

If the read-under-write flag is set to new, then a read port always re-
turns the value existing in the memory on the same cycle that the read was
made available. Intuitively, this is modeled as a combinational read from the
memory after delaying the read address by the appropriate read latency.

If the read-under-write flag is set to undefined, then the value held by
the read port after the appropriate read latency is undefined.

In all cases, if a memory location is written to by more than one port on
the same cycle, the stored value is undefined.

5.11 Instances

FIRRTL modules are instantiated with the instance statement. The follow-
ing example demonstrates creating an instance named myinstance of the
MyModule module within the top level module Top.

circuit Top :

module MyModule :

input a: UInt

output b: UInt

b <= a

module Top :

inst myinstance of MyModule

The resulting instance has a bundle type. Each port of the instantiated
module is represented by a field in the bundle with the same name and type as
the port. The fields corresponding to input ports are flipped to indicate their

data flows in the opposite direction as the output ports. The myinstance

instance in the example above has type {flip a:UInt, b:UInt}.
Modules have the property that instances can always be inlined into the

parent module without affecting the semantics of the circuit.
To disallow infinitely recursive hardware, modules cannot contain in-

stances of itself, either directly, or indirectly through instances of other mod-
ules it instantiates.

5.12 Stops

The stop statement is used to halt simulations of the circuit. Backends
are free to generate hardware to stop a running circuit for the purpose of
debugging, but this is not required by the FIRRTL specification.

A stop statement requires a clock signal, a halt condition signal that has
a single bit unsigned integer type, and an integer exit code.

wire clk:Clock

wire halt:UInt<1>

stop(clk,halt,42)

...

5.13 Formatted Prints

The formatted print statement is used to print a formatted string during
simulations of the circuit. Backends are free to generate hardware that relays
this information to a hardware test harness, but this is not required by the
FIRRTL specification.

A printf statement requires a clock signal, a print condition signal, a
format string, and a variable list of argument signals. The condition signal
must be a single bit unsigned integer type, and the argument signals must
each have a ground type.

wire clk:Clock

wire condition:UInt<1>

wire a:UInt

wire b:UInt

printf(clk, condition, "a in hex: %x, b in decimal:%d.\n", a, b)

...

On each positive clock edge, when the condition signal is high, the printf
statement prints out the format string where its argument placeholders are
substituted with the value of the corresponding argument.

5.13.1 Format Strings

Format strings support the following argument placeholders:

• %b : Prints the argument in binary

• %d : Prints the argument in decimal

• %x : Prints the argument in hexadecimal

• %% : Prints a single % character

Format strings support the following escape characters:

• \n : New line

• \t : Tab

• \\ : Back slash

• \" : Double quote

• \’ : Single quote

6 Expressions

FIRRTL expressions are used for creating literal unsigned and signed inte-
gers, for referring to a declared circuit component, for statically and dynami-
cally accessing a nested element within a component, for creating multiplex-
ors and conditionally valid signals, and for performing primitive operations.

6.1 Unsigned Integers

A literal unsigned integer can be created given a non-negative integer value
and an optional positive bit width. The following example creates a 10-bit
unsigned integer representing the number 42.

UInt<10>(42)

Note that it is an error to supply a bit width that is not large enough to
fit the given value. If the bit width is omitted, then the minimum number of
bits necessary to fit the given value will be inferred.

UInt(42)

6.2 Unsigned Integers from Literal Bits

A literal unsigned integer can alternatively be created given a string repre-
senting its bit representation and an optional bit width.

The following radices are supported:

1. 0b : For representing binary numbers.

2. 0o : For representing octal numbers.

3. 0x : For representing hexadecimal numbers.

If a bit width is not given, the number of bits in the bit representation
is directly represented by the string. The following examples create a 8-bit
integer representing the number 13.

UBits("0b00001101")

UBits("0x0D")

If the provided bit width is larger than the number of bits represented by
the string, then the resulting value is equivalent to the string zero-extended
up to the provided bit width. If the provided bit width is smaller than the
number of bits represented by the string, then the resulting value is equivalent
to the string truncated down to the provided bit width. All truncated bits
must be zero.

The following examples create a 7-bit integer representing the number 13.

UBits<7>("0b00001101")

UBits<7>("0o015")

UBits<7>("0xD")

6.3 Signed Integers

Similar to unsigned integers, a literal signed integer can be created given
an integer value and an optional positive bit width. The following example
creates a 10-bit unsigned integer representing the number -42.

SInt<10>(-42)

Note that it is an error to supply a bit width that is not large enough to
fit the given value using two’s complement representation. If the bit width is
omitted, then the minimum number of bits necessary to fit the given value
will be inferred.

SInt(-42)

6.4 Signed Integers from Literal Bits

Similar to unsigned integers, a literal signed integer can alternatively be
created given a string representing its bit representation and an optional bit
width.

If a bit width is not given, the number of bits in the bit representation
is directly represented by the string. The following examples create a 8-bit
integer representing the number -13.

SBits("0b11110011")

SBits("0xF3")

If the provided bit width is larger than the number of bits represented by
the string, then the resulting value is equivalent to the string sign-extended
up to the provided bit width. If the provided bit width is smaller than the
number of bits represented by the string, then the resulting value is equivalent
to the string truncated down to the provided bit width. All truncated bits
must match the sign bit of the final truncated number.

The following examples create a 7-bit integer representing the number
-13.

SBits<7>("0b10011")

SBits<7>("0o763")

SBits<7>("0xF3")

6.5 References

A reference is simply a name that refers to a previously declared circuit
component. It may refer to a module port, node, wire, register, instance, or
memory.

The following example connects a reference expression in, referring to the
previously declared port in, to the reference expression out, referring to the
previously declared port out.

module MyModule :

input in: UInt

output out: UInt

out <= in

In the rest of the document, for brevity, the names of components will be
used to refer to a reference expression to that component. Thus, the above
example will be rewritten as “the port in is connected to the port out”.

6.6 Subfields

The subfield expression refers to a subelement of an expression with a bundle
type.

The following example connects the in port to the a subelement of the
out port.

module MyModule :

input in: UInt

output out: {a:UInt, b:UInt}

out.a <= in

6.7 Subindices

The subindex expression statically refers, by index, to a subelement of an
expression with a vector type. The index must be a non-negative integer and
cannot be equal to or exceed the length of the vector it indexes.

The following example connects the in port to the fifth subelement of the
out port.

module MyModule :

input in: UInt

output out: UInt[10]

out[4] <= in

6.8 Subaccesses

The subaccess expression dynamically refers to a subelement of a vector-
typed expression using a calculated index. The index must be an expression
with an unsigned integer type.

The following example connects the n’th subelement of the in port to the
out port.

module MyModule :

input in: UInt[3]

input n: UInt<2>

output out: UInt

out <= in[n]

A connection from a subaccess expression can be modeled by conditionally
connecting from every subelement in the vector, where the condition holds
when the dynamic index is equal to the subelement’s static index.

module MyModule :

input in: UInt[3]

input n: UInt<2>

output out: UInt

when eq(n, UInt(0)) :

out <= in[0]

else when eq(n, UInt(1)) :

out <= in[1]

else when eq(n, UInt(2)) :

out <= in[2]

else :

out is invalid

The following example connects the in port to the n’th subelement of
the out port. All other subelements of the out port are connected from the
corresponding subelements of the default port.

36



Specification for the FIRRTL Language Version 0.2.0

module MyModule :

input in: UInt

input default: UInt[3]

input n: UInt<2>

output out: UInt[3]

out <= default

out[n] <= in

A connection to a subaccess expression can be modeled by conditionally
connecting to every subelement in the vector, where the condition holds when
the dynamic index is equal to the subelement’s static index.

module MyModule :

input in: UInt

input default: UInt[3]

input n: UInt<2>

output out: UInt[3]

out <= default

when eq(n, UInt(0)) :

out[0] <= in

else when eq(n, UInt(1)) :

out[1] <= in

else when eq(n, UInt(2)) :

out[2] <= in

The following example connects the in port to the m’th UInt subelement
of the n’th vector-typed subelement of the out port. All other subelements
of the out port are connected from the corresponding subelements of the
default port.

module MyModule :

input in: UInt

input default: UInt[2][2]

input n: UInt<1>

input m: UInt<1>

output out: UInt[2][2]

out <= default

out[n][m] <= in

A connection to an expression containing multiple nested subaccess ex-
pressions can also be modeled by conditionally connecting to every subele-
ment in the expression. However the condition holds only when all dynamic
indices are equal to all of the subelement’s static indices.

module MyModule :

input in: UInt

input default: UInt[2][2]

input n: UInt<1>

input m: UInt<1>

output out: UInt[2][2]

out <= default

when and(eq(n, UInt(0)), eq(m, UInt(0))) :

out[0][0] <= in

else when and(eq(n, UInt(0)), eq(m, UInt(1))) :

out[0][1] <= in

else when and(eq(n, UInt(1)), eq(m, UInt(0))) :

out[1][0] <= in

else when and(eq(n, UInt(1)), eq(m, UInt(1))) :

out[1][1] <= in

6.9 Multiplexors

A multiplexor outputs one of two input expressions depending on the value
of an unsigned single bit selection signal.

The following example connects to the c port the result of selecting be-
tween the a and b ports. The a port is selected when the sel signal is high,
otherwise the b port is selected.

module MyModule :

input a: UInt

input b: UInt

input sel: UInt<1>

output c: UInt

c <= mux(sel, a, b)

A multiplexor expression is legal only if the following holds.

1. The type of the selection signal is a single bit unsigned integer.

2. The types of the two input expressions are equivalent.

3. The types of the two input expressions are passive (see section 4.4).

6.10 Conditionally Valids

A conditionally valid expression is expressed as an input expression guarded
with an unsigned single bit valid signal. It outputs the input expression when
the valid signal is high, otherwise the result is undefined.

The following example connects the a port to the c port when the valid

signal is high. Otherwise, the value of the c port is undefined.

module MyModule :

input a: UInt

input valid: UInt<1>

output c: UInt

c <= validif(valid, a)

A conditionally valid expression is legal only if the following holds.

1. The type of the valid signal is a single bit unsigned integer.

2. The type of the input expression is passive (see section 4.4).

Conditional statements can be equivalently expressed as multiplexors and
conditionally valid expressions. See section 5.9 for details.

6.11 Primitive Operations

All fundamental operations on ground types are expressed as a FIRRTL prim-
itive operation. In general, each operation takes some number of argument
expressions, along with some number of static integer literal parameters.

The general form of a primitive operation is expressed as follows:

op(arg0, arg1, ..., argn, int0, int1, ..., intm)

The following examples of primitive operations demonstrate adding two
expressions, a and b, shifting expression a left by 3 bits, selecting the fourth
bit through and including the seventh bit in the a expression, and interpreting
the expression x as a Clock typed signal.

add(a, b)

shl(a, 3)

bits(a, 7, 4)

asClock(x)

Section 7 will describe the format and semantics of each primitive oper-
ation.

7 Primitive Operations

The arguments of all primitive operations must be expressions with ground
types, while their parameters are static integer literals. Each specific op-
eration can place additional restrictions on the number and types of their
arguments and parameters.

Notationally, the width of an argument e is represented as we.

7.1 Add Operation

Name Arguments Parameters Arg Types Result Type Result Width

add (e1,e2) () (UInt,UInt) UInt max(we1,we2)+1

(UInt,SInt) SInt max(we1,we2)+1

(SInt,UInt) SInt max(we1,we2)+1

(SInt,SInt) SInt max(we1,we2)+1

The add operation result is the sum of e1 and e2 without loss of precision.

7.2 Subtract Operation

Name Arguments Parameters Arg Types Result Type Result Width

sub (e1,e2) () (UInt,UInt) SInt max(we1,we2)+1

(UInt,SInt) SInt max(we1,we2)+1

(SInt,UInt) SInt max(we1,we2)+1

(SInt,SInt) SInt max(we1,we2)+1

The subtract operation result is e2 subtracted from e1, without loss of pre-
cision.

7.3 Multiply Operation

Name Arguments Parameters Arg Types Result Type Result Width

mul (e1,e2) () (UInt,UInt) UInt we1+we2
(UInt,SInt) SInt we1+we2
(SInt,UInt) SInt we1+we2
(SInt,SInt) SInt we1+we2

The multiply operation result is the product of e1 and e2, without loss of
precision.

7.4 Divide Operation

Name Arguments Parameters Arg Types Result Type Result Width

div (num,den) () (UInt,UInt) UInt wnum
(UInt,SInt) SInt wnum+1

(SInt,UInt) SInt wnum
(SInt,SInt) SInt wnum+1

The divide operation divides num by den, truncating the fractional portion
of the result. This is equivalent to rounding the result towards zero.

7.5 Modulus Operation

Name Arguments Parameters Arg Types Result Type Result Width

mod (num,den) () (UInt,UInt) UInt min(wnum,wden)

(UInt,SInt) UInt min(wnum,wden)

(SInt,UInt) SInt min(wnum,wden+1)

(SInt,SInt) SInt min(wnum,wden)

The modulus operation yields the remainder from dividing num by den, keep-
ing the sign of the numerator. Together with the divide operator, the mod-
ulus operator satisfies the relationship below:

num = add(mul(den,div(num,den)),mod(num,den))}

7.6 Comparison Operations

Name Arguments Parameters Arg Types Result Type Result Width

lt,leq,gt, (e1,e2) () (UInt,UInt) UInt 1

geq,eq,neq (UInt,SInt) UInt 1

(SInt,UInt) UInt 1

(SInt,SInt) UInt 1

The comparison operations return an unsigned 1 bit signal with value one if
e1 is less than (lt), less than or equal to (leq), greater than (gt), greater
than or equal to (geq), equal to (eq), or not equal to (neq) e2. The operation
returns a value of zero otherwise.

7.7 Padding Operations

Name Arguments Parameters Arg Types Result Type Result Width

pad (e) (n) (UInt) UInt max(we,n)

(SInt) SInt max(we,n)

If e’s bit width is smaller than n, then the pad operation zero-extends or
sign-extends e up to the given width n. Otherwise, the result is simply e. n

must be non-negative.

7.8 Interpret As UInt

Name Arguments Parameters Arg Types Result Type Result Width

asUInt (e) () (UInt) UInt we
(SInt) UInt we
(Clock) UInt 1

The interpret as UInt operation reinterprets e’s bits as an unsigned integer.

7.9 Interpret As SInt

Name Arguments Parameters Arg Types Result Type Result Width

asSInt (e) () (UInt) SInt we
(SInt) SInt we
(Clock) SInt 1

The interpret as SInt operation reinterprets e’s bits as a signed integer ac-
cording to two’s complement representation.

7.10 Interpret as Clock

Name Arguments Parameters Arg Types Result Type Result Width

asClock (e) () (UInt) Clock n/a

(SInt) Clock n/a

(Clock) Clock n/a

The result of the interpret as clock operation is the Clock typed signal ob-
tained from interpreting a single bit integer as a clock signal.

7.11 Shift Left Operation

Name Arguments Parameters Arg Types Result Type Result Width

shl (e) (n) (UInt) UInt we+n

(SInt) SInt we+n

The shift left operation concatenates n zero bits to the least significant end
of e. n must be non-negative.

7.12 Shift Right Operation

Name Arguments Parameters Arg Types Result Type Result Width

shr (e) (n) (UInt) UInt we-n

(SInt) SInt we-n

The shift right operation truncates the least significant n bits from e. n must
be non-negative and strictly less than the bit width of e.

7.13 Dynamic Shift Left Operation

Name Arguments Parameters Arg Types Result Type Result Width

dshl (e, n) () (UInt, UInt) UInt we + 2^wn
(SInt, UInt) SInt we + 2^wn

The dynamic shift left operation shifts the bits in e n places towards the
most significant bit. n zeroes are shifted in to the least significant bits.

7.14 Dynamic Shift Right Operation

Name Arguments Parameters Arg Types Result Type Result Width

dshr (e, n) () (UInt, UInt) UInt we
(SInt, UInt) SInt we

The dynamic shift right operation shifts the bits in e n places towards the
least significant bit. n signed or zeroed bits are shifted in to the most signif-
icant bits, and the n least significant bits are truncated.

7.15 Arithmetic Convert to Signed Operation

Name Arguments Parameters Arg Types Result Type Result Width

cvt (e) () (UInt) SInt we+1

(SInt) SInt we

The result of the arithmetic convert to signed operation is a signed integer
representing the same numerical value as e.

7.16 Negate Operation

Name Arguments Parameters Arg Types Result Type Result Width

neg (e) () (UInt) SInt we+1

(SInt) SInt we+1

The result of the negate operation is a signed integer representing the negated
numerical value of e.

7.17 Bitwise Complement Operation

Name Arguments Parameters Arg Types Result Type Result Width

not (e) () (UInt) UInt we
(SInt) UInt we

The bitwise complement operation performs a logical not on each bit in e.

7.18 Binary Bitwise Operations

Name Arguments Parameters Arg Types Result Type Result Width

and,or,xor (e1, e2) () (UInt,UInt) UInt max(we1,we2)

(UInt,SInt) UInt max(we1,we2)

(SInt,UInt) UInt max(we1,we2)

(SInt,SInt) UInt max(we1,we2)

The above bitwise operations perform a bitwise and, or, or exclusive or on
e1 and e2. The result has the same width as its widest argument, and
any narrower arguments are automatically zero-extended or sign-extended
to match the width of the result before performing the operation.

7.19 Bitwise Reduction Operations

Name Arguments Parameters Arg Types Result Type Result Width

andr,orr,xorr (e) () (UInt) UInt 1

(SInt) UInt 1

The bitwise reduction operations correspond to a bitwise and, or, and exclu-
sive or operation, reduced over every bit in e.

7.20 Concatenate Operation

Name Arguments Parameters Arg Types Result Type Result Width

cat (e1,e2) () (UInt, UInt) UInt we1+we2
(UInt, SInt) UInt we1+we2
(SInt, UInt) UInt we1+we2
(SInt, SInt) UInt we1+we2

The result of the concatenate operation is the bits of e1 concatenated to the
most significant end of the bits of e2.

7.21 Bit Extraction Operation

Name Arguments Parameters Arg Types Result Type Result Width

bits (e) (hi,lo) (UInt) UInt hi-lo+1

(SInt) UInt hi-lo+1

The result of the bit extraction operation are the bits of e between lo (in-
clusive) and hi (inclusive). hi must be greater than or equal to lo. Both hi

and lo must be non-negative and strictly less than the bit width of e.

7.22 Head

Name Arguments Parameters Arg Types Result Type Result Width

head (e) (n) (UInt) UInt n

(SInt) UInt n

The result of the head operation are the n most significant bits of e. n must
be positive and less than or equal to the bit width of e.

7.23 Tail

Name Arguments Parameters Arg Types Result Type Result Width

tail (e) (n) (UInt) UInt we-n

(SInt) UInt we-n

The tail operation truncates the n most significant bits from e. n must be
non-negative and strictly less than the bit width of e.

8 Genders

An expression’s gender partially determines the legality of connecting to and
from the expression. Every expression is classified as either male, female, or
bi-gender. For details on connection rules refer back to sections 5.1 and 5.2.

The gender of a reference to a declared circuit component depends on
the kind of circuit component. A reference to an input port, an instance, a
memory, and a node, is male. A reference to an output port is female. A
reference to a wire or register is bi-gender.

The gender of a subindex or subaccess expression is the gender of the
vector-typed expression it indexes or accesses.

The gender of a subfield expression depends upon the orientation of the
field. If the field is not flipped, its gender is the same gender as the bundle-
typed expression it selects its field from. If the field is flipped, then its gender
is the reverse of the gender of the bundle-typed expression it selects its field
from. The reverse of male is female, and vice-versa. The reverse of bi-gender
remains bi-gender.

The gender of all other expressions are male.

9 Width Inference

For all circuit components declared with unspecified widths, the FIRRTL
compiler will infer the minimum possible width that maintains the legality
of all its incoming connections. If a component has no incoming connections,
and the width is unspecified, then an error is thrown to indicate that the
width could not be inferred.

For module input ports with unspecified widths, the inferred width is the
minimum possible width that maintains the legality of all incoming connec-
tions to all instantiations of the module.

The width of a ground-typed multiplexor expression is the maximum
of its two corresponding input widths. For multiplexing aggregate-typed
expressions, the resulting widths of each leaf subelement is the maximum of
its corresponding two input leaf subelement widths.

The width of a conditionally valid expression is the width of its input
expression.

The width of each primitive operation is detailed in section 7.
The width of the integer literal expressions is detailed in their respective

sections.

10 Namespaces

All modules in a circuit exist in the same module namespace, and thus must
all have a unique name.

Each module has an identifier namespace containing the names of all port
and circuit component declarations. Thus, all declarations within a module
must have unique names. Furthermore, the set of component declarations
within a module must be prefix unique. Please see section 10.2 for the defi-
nition of prefix uniqueness.

Within a bundle type declaration, all field names must be unique.
Within a memory declaration, all port names must be unique.
During the lowering transformation, all circuit component declarations

with aggregate types are rewritten as a group of component declarations,
each with a ground type. The name expansion algorithm in section 10.1
calculates the names of all replacement components derived from the original
aggregate-typed component.

After the lowering transformation, the names of the lowered circuit com-
ponents are guaranteed by the name expansion algorithm and thus can be re-
liably referenced by users to pair meta-data or other annotations with named
circuit components.

10.1 Name Expansion Algorithm

Given a component with a ground type, the name of the component is re-
turned.

Given a component with a vector type, the suffix $i is appended to the
expanded names of each subelement, where i is the index of each subelement.

Given a component with a bundle type, the suffix $f is appended to
the expanded names of each subelement, where f is the field name of each
subelement.

10.2 Prefix Uniqueness

The symbol sequence of a name is the ordered list of strings that results from
splitting the name at each occurence of the ‘$’ character.

A symbol sequence a is a prefix of another symbol sequence b if the strings
in a occur in the beginning of b.

48



Specification for the FIRRTL Language Version 0.2.0

A set of names are defined to be prefix unique if there exists no two names
such that the symbol sequence of one is a prefix of the symbol sequence of
the other.

As an example firetruck$y$z shares a prefix with firetruck$y and
firetruck, but does not share a prefix with fire.

11 The Lowered FIRRTL Form

The lowered FIRRTL form, LoFIRRTL, is a restricted subset of the FIRRTL
language that omits many of the higher level constructs. All conformant
FIRRTL compilers must provide a lowering transformation that transforms
arbitrary FIRRTL circuits into equivalent LoFIRRTL circuits.

A FIRRTL circuit is defined to be a valid LoFIRRTL circuit if it obeys
the following restrictions:

• All components must be declared with a ground type and explicit
widths.

• The partial connect statement is not used.

• The conditional statement is not used.

• All components are connected to exactly once.

The additional restrictions give LoFIRRTL a direct correspondence to a
circuit netlist.

Low level circuit transformations can be conveniently written by first
lowering a circuit to its LoFIRRTL form, then operating on the restricted
(and thus simpler) subset of constructs. Note that circuit transformations
are still free to generate high level constructs as they can simply be lowered
again.

The following module:

module MyModule :

input in: {a:UInt<1>, b:UInt<2>[3]}

input clk: Clock

output out: UInt

wire c: UInt

c <= in.a

reg r: UInt[3], clk

r <= in.b

when c :

r[1] <= in.a

out <= r[0]

is rewritten as the following equivalent LoFIRRTL circuit by the lowering
transform.

module MyModule :

input in$a: UInt<1>

input in$b$0: UInt<2>

input in$b$1: UInt<2>

input in$b$2: UInt<2>

input clk: Clock

output out: UInt<2>

wire c: UInt<1>

c <= in$a

reg r$0: UInt<2>, clk

reg r$1: UInt<2>, clk

reg r$2: UInt<2>, clk

r$0 <= in$b$0

r$1 <= mux(c, in$a, in$b$1)

r$2 <= in$b$2

out <= r$0

12 Details about Syntax

FIRRTL’s syntax is designed to be human-readable but easily algorithmically
parsed.

The following characters are allowed in identifiers: upper and lower case
letters, digits, as well as the punctuation characters ~!@#$%^*-_+=?/. Iden-
tifiers cannot begin with a digit.

An integer literal in FIRRTL begins with one of the following, where ‘#’
represents a digit between 0 and 9.

• ‘0x’ : For indicating a hexadecimal number. The rest of the literal must
consist of either digits or a letter between ‘A’ and ‘F’, or the separator
‘ ’.

• ‘0o’ : For indicating an octal number. The rest of the literal must
consist of digits between 0 and 7, or the separator ‘ ’.

• ‘0b’ : For indicating a binary number. The rest of the literal must
consist of either 0 or 1, or the separator ‘ ’.

• ‘-#’ : For indicating a negative decimal number. The rest of the literal
must consist of digits between 0 and 9.

• ‘#’ : For indicating a positive decimal number. The rest of the literal
must consist of digits between 0 and 9.

Comments begin with a semicolon and extend until the end of the line.
Commas are treated as whitespace, and may be used by the user for clarity
if desired.

Block structuring is indicated using indentation. Statements are com-
bined into statement groups by surrounding them with parenthesis. A colon
at the end of a line will automatically surround the next indented region
with parenthesis and thus create a statement group.

The following statement:

when c :

a <= b

else :

c <= d

e <= f

can be equivalently expressed on a single line as follows.

when c : (a <= b) else : (c <= d, e <= f)

All circuits, modules, ports and statements can optionally be preceded
with the info token @["filename", line, col] to annotate them with the
source file information from where they were generated.

The following example shows the info tokens included:

@["myfile.txt" 14, 8] circuit Top :

@["myfile.txt" 15, 2] module Top :

@["myfile.txt" 16, 3] output out:UInt

@["myfile.txt" 17, 3] input b:UInt<32>

@["myfile.txt" 18, 3] input c:UInt<1>

@["myfile.txt" 19, 3] input d:UInt<16>

@["myfile.txt" 21, 8] wire a:UInt

@["myfile.txt" 24, 8] when c :

@["myfile.txt" 27, 16] a <= b

else :

@["myfile.txt" 29, 17] a <= d

@["myfile.txt" 34, 4] out <= add(a,a)

13 FIRRTL Language Definition

13.1 Notation

The concrete syntax of FIRRTL is defined in section 13.2. Productions in
the syntax tree are italicized and keywords are written in monospaced font.
The special productions id, int, and string, indicates an identifier, an integer
literal, and a string respectively. The notation JeK... is used to indicate that
e is repeated zero or more times, and the notation JeK? is used to indicate
that including e is optional.

13.2 Concrete Syntax Tree

circuit = JinfoK? circuit id : (JmoduleK...) Circuit
module = JinfoK? module id : (JportK... stmt) Module

| JinfoK? extmodule id : (JportK...) External Module
port = JinfoK? dir id : type Port
dir = input | output Port Direction

type = UIntJ<int>K? Unsigned Integer
| SIntJ<int>K? Signed Integer
| Clock Clock
| {JfieldK...} Bundle
| type[int] Vector

field = JflipK? id : type Bundle Field
stmt = JinfoK? wire id : type Wire

| JinfoK? reg id : type, exp Jexp, expK? Register
| JinfoK? mem id : ( Memory

data-type => type
depth => int
read-latency => int
write-latency => int
read-under-write => ruw
Jreader => idK...
Jwriter => idK...
Jreadwriter => idK...)

| JinfoK? inst id of id Instance
| JinfoK? node id = exp Node
| JinfoK? exp <= exp Connect
| JinfoK? exp <- exp Partial Connect
| JinfoK? exp is invalid Invalidate
| JinfoK? when exp : stmt Jelse : stmtK? Conditional
| JinfoK? stop(exp, exp, int) Stop
| JinfoK? printf(exp, exp, string, JexpK...) Printf
| JinfoK? skip Empty
| JinfoK? (JstmtK...) Statement Group

ruw = old | new | undefined Read Under Write Flag
info = @[string, int, int] File Information Token

exp = UIntJ<int>K?(int) Literal Unsigned Integer
| UIntJ<int>K?(string) Literal Unsigned Integer From Bits
| SIntJ<int>K?(int) Literal Signed Integer
| SIntJ<int>K?(string) Literal Signed Integer From Bits
| id Reference
| exp.id Subfield
| exp[int] Subindex
| exp[exp] Subaccess
| mux(exp, exp, exp) Multiplexor
| validif(exp, exp) Conditionally Valid
| primop(JexpK..., JintK...) Primitive Operation

primop = add Add
| sub Subtract
| mul Multiply
| div Divide
| mod Modulo
| lt Less Than
| leq Less or Equal
| gt Greater Than
| geq Greater or Equal
| eq Equal
| neq Not-Equal
| pad Pad
| asUInt Interpret Bits as UInt
| asSInt Interpret Bits as SInt
| asClock Interpret as Clock
| shl Shift Left
| shr Shift Right
| dshl Dynamic Shift Left
| dshr Dynamic Shift Right
| cvt Arithmetic Convert to Signed
| neg Negate
| not Not
| and And
| or Or
| xor Xor
| andr And Reduce
| orr Or Reduce
| xorr Xor Reduce
| cat Concatenation
| bits Bit Extraction
| head Head
| tail Tail

55


	Introduction
	Background
	Design Philosophy

	Acknowledgements
	Circuits and Modules
	Circuits
	Modules
	Externally Defined Modules

	Types
	Ground Types
	Integer Types
	Clock Type

	Vector Types
	Bundle Types
	Passive Types
	Type Equivalence
	Weak Type Equivalence
	Oriented Types
	Conversion to Oriented Types
	Oriented Type Equivalence


	Statements
	Connects
	The Connection Algorithm

	Partial Connects
	The Partial Connection Algorithm

	Statement Groups
	Last Connect Semantics

	Empty
	Wires
	Registers
	Invalidates
	The Invalidate Algorithm

	Nodes
	Conditionals
	Syntactic Shorthands
	Nested Declarations
	Initialization Coverage
	Scoping
	Conditional Last Connect Semantics

	Memories
	Read Ports
	Write Ports
	Readwrite Ports
	Read Under Write Behaviour

	Instances
	Stops
	Formatted Prints
	Format Strings


	Expressions
	Unsigned Integers
	Unsigned Integers from Literal Bits
	Signed Integers
	Signed Integers from Literal Bits
	References
	Subfields
	Subindices
	Subaccesses
	Multiplexors
	Conditionally Valids
	Primitive Operations

	Primitive Operations
	Add Operation
	Subtract Operation
	Multiply Operation
	Divide Operation
	Modulus Operation
	Comparison Operations
	Padding Operations
	Interpret As UInt
	Interpret As SInt
	Interpret as Clock
	Shift Left Operation
	Shift Right Operation
	Dynamic Shift Left Operation
	Dynamic Shift Right Operation
	Arithmetic Convert to Signed Operation
	Negate Operation
	Bitwise Complement Operation
	Binary Bitwise Operations
	Bitwise Reduction Operations
	Concatenate Operation
	Bit Extraction Operation
	Head
	Tail

	Genders
	Width Inference
	Namespaces
	Name Expansion Algorithm
	Prefix Uniqueness

	The Lowered FIRRTL Form
	Details about Syntax
	FIRRTL Language Definition
	Notation
	Concrete Syntax Tree


